
HYBRID SEQUENCER VERSION 2.0 

Overview 
The Hybrid Sequencer is a program designed to communicate with the FlexTek  FCIF010 
FlexBus Board  via a RS232 serial port from a PC.  The FlexBus in turn controls a 
custom interface board containing MOSFET power switches that activate the igniter and 
solenoids of the hybrid motor static test stand.  
 

 
 
The FC1F010 is a multi-function micro controller with a modest amount of Digital I/O 
and Analog to Digital converters 
 

http://www.flex-tek.com/index.html 
 

 
Figure 1 FCIF010..........FlexBusTM Interface Board  

The FlexBus board has 5 A/D channels, 4 digital outputs, 4 digital inputs, 1 Pulse Width 
modulated output and 1 timer. Currently Hybrid Sequencer 2.0 only utilizes 4 digital 
outputs. 



The  FlexBus micro controller is designed to be easily programmed with Visual Basic 
and utilizes an ActiveX component named EZX.ocx (“FTView” for newer versions) 
which handles the interface between the Visual Basic program and the micro controller 
hardware. This ActiveX control can be downloaded from the FlexTek website and can be 
inserted as a Control Component in your Visual Basic project. 

 

Figure 2 EZX 1.0 ActiveX control 

The Hybrid Sequencer 2.0 program is designed to command this ActiveX control in a 
particular timed sequence that is determined by the user through a graphical interface. 

Installation 
 
To install Hybrid Sequencer 2.0: 

1) Download appropriate installation file. (Win98, 2000 or XP) 
2) Click on Hybrid Sequencer Setup file. 
3) Select Destination location from installed files. 
4) After installation, start program by clicking on desktop icon or go to START > 

PROGRAMS. 

 



Operation 

Upon starting the Hybrid Sequencer 2.0 executable file the ‘FlexControl’ form is loaded.  
This custom interface is what handles the automated control of the EZX 1.0 ActiveX 
object.  

The Hybrid Sequencer program has two primary functions: 

1) Automated control the digital output channels of the FlexBus and Interface board 
to activate and deactivate them at the appropriate times thus turning on and off the 
solenoids for the oxidizer, purge and igniter functions. 

2) Serve as a manual interface with the FlexBus and Interface board by selecting the 
digital output channels as required through the ‘Test Panel’ buttons. 

 

Interface 

1) Connect:   In order for the program to communicate with the FlexBus Interface 
board the  “Connect” button must be pushed.  This button sets EZX1.CommPort = 1 
which establishes serial communication with the FlexBus board on Comm Port 1. 

2) Countdown:   The number of seconds that will be counted down before the 
igniter is started.  Countdown must have a minimum value of 1 second. 



3) Ignition:    The time at which the igniter becomes active by setting EZX1.DOKstate 
= 1 the value is fixed at T minus 0 and cannot be changed.  This is the time point 
on which the rest of the firing sequence is referenced. 

4) Oxidizer ON:   The time at which the solenoid controlling the pressurized 
nitrogen, to the OPEN side of the oxidizer pneumatic ball valve, is activated by 
setting EZX1.DOSstate = 1 at this point the igniter is de asserted by setting 
EZX1.DOKstate = 0.  Oxidizer ON must be at least one second later than Ignition.  

5) Oxidizer OFF:   The time at which the solenoid controlling the pressurized 
nitrogen, to the CLOSE side of the oxidizer pneumatic ball valve, is activated by 
setting EZX1.DOEstate = 1. Also at this point the solenoid controlling the OPEN 
side is deactivated by setting EZX1.DOSstate = 0.If no purge is used then the 
CLOSE solenoid will be deactivated by setting EZX1.DOEstate = 0 two seconds 
after Oxidizer OFF.  Oxidizer OFF must occur at least 1 second after Oxidizer 
ON. 

6) Purge ON:   If a purge is used then this is the time at which the inert gas purge 
solenoid is activated by setting the EZX1.DONstate = 1. Also at this point the 
solenoid controlling the CLOSE side of the oxidizer pneumatic ball valve is de 
activated by setting EZX1.DOEstate = 0. Purge ON must occur at least 1 second 
after Oxidizer OFF. 

7) Purge OFF:   If a purge is used then this is the time at which the inert gas purge 
solenoid is deactivated by setting the EZX1.DONstate = 1. Purge OFF must occur at 
least 1 second after Purge ON. 

8) Status:   This is the general message field that displays status messages as well as 
error messages. 

9) START:   This button will be enabled once the connect button is pressed. Once 
the Start button is pressed a logic check is performed on the sequence time fields 
to make sure they are entered correctly.  If the sequence logic is valid then the 
countdown will proceed. 

10) ABORT:   Terminates the sequence and deactivates DOK, DOS, DOE and DON. 

 

Test Panel 

11) Igniter:    Manually toggles DOK activate/deactivate when clicked. 
12) Oxidizer ON:   Manually toggles DOS activate/deactivate when clicked. 
13) Oxidizer OFF:   Manually toggles DOE activate/deactivate when clicked. 
14) Purge:   Manually toggles DON activate/deactivate when clicked. 

NOTE 

?? If the Hybrid Sequencer 2.0 ‘FlexControl’ form is maximized on the screen, 
the ‘show flextek interface’ button will become visible. Clicking this button 
will allow the user to view the activity and status of the control signals being 
sent to the micro controller from the executable via this ActiveX control. 



 

 

 

Figure 3 PSAS custom hybrid interface board 



 

Figure 4 FlexBusTM Board Block Diagram 

 
 
 
 

Visual Basic 6.0 Source Code for Hybrid Sequencer Version 2.0 
 
Dim COUNTDOWN As Integer 
Dim COUNTUP As Integer 
Dim OXIDIZERON As Integer 
Dim OXIDIZEROFF As Integer 
Dim PURGEON As Integer 
Dim PURGEOFF As Integer 
Dim ERROR As Integer 
Dim PURGE As Boolean 
Dim SequenceCheck As Boolean 
Dim COUNTDOWNACTIVE As Boolean 
 
Private Sub cmdabort_Click() 
    EZX1.DOKstate = 0 
    EZX1.DOEstate = 0 
    EZX1.DONstate = 0 
    EZX1.DOSstate = 0 
    txtcount.Text = txtcountdown.Text 
    lbltsign.Text = "T minus" 
    tmrhybrid1.Enabled = False 
    txtstatus.Text = "Sequence Aborted" 
    EZX1.UpdateCmd = True 
End Sub 
 
Private Sub cmdconnect_Click() 
    If EZX1.CommPort = 0 Then 
        EZX1.CommPort = 1 
        txtstatus.Text = "Connected" 
        EZX1.UpdateCmd = True 
        cmdstart.Enabled = True 
    ElseIf EZX1.CommPort = 1 Then 
        EZX1.CommPort = 0 
        txtstatus.Text = "Disconnected" 
        cmdstart.Enabled = True 



    End If 
End Sub 
 
Private Sub cmdigniter_Click() 
    If EZX1.DOKstate = 0 Then 
        EZX1.DOKstate = 1 
        txtstatus.Text = "DOK Active" 
    ElseIf EZX1.DOKstate = 1 Then 
        EZX1.DOKstate = 0 
        txtstatus.Text = "DOK Inactive" 
    End If 
    EZX1.UpdateCmd = True 
End Sub 
 
Private Sub cmdoxidizeroff_Click() 
    If EZX1.DOEstate = 0 Then 
        EZX1.DOEstate = 1 
        txtstatus.Text = "DOE Active" 
    ElseIf EZX1.DOEstate = 1 Then 
        EZX1.DOEstate = 0 
        txtstatus.Text = "DOE Inactive" 
    End If 
    EZX1.UpdateCmd = True 
End Sub 
 
Private Sub cmdoxidizeron_Click() 
    If EZX1.DOSstate = 0 Then 
        EZX1.DOSstate = 1 
        txtstatus.Text = "DOS Active" 
    ElseIf EZX1.DOSstate = 1 Then 
        EZX1.DOSstate = 0 
        txtstatus.Text = "DOS Inactive" 
    End If 
        EZX1.UpdateCmd = True 
End Sub 
 
Private Sub cmdpurge_Click() 
    If EZX1.DONstate = 0 Then 
        EZX1.DONstate = 1 
        txtstatus.Text = "DON Active" 
    ElseIf EZX1.DONstate = 1 Then 
        EZX1.DONstate = 0 
        txtstatus.Text = "DON Inactive" 
    End If 
        EZX1.UpdateCmd = True 
End Sub 
 
Private Sub cmdstart_Click() 
    COUNTDOWN = Val(txtcountdown.Text) 
    COUNTUP = Val(txtignition.Text) 
    COUNTDOWNACTIVE = True 
    OXIDIZERON = Val(txtoxidizeron.Text) 
    OXIDIZEROFF = Val(txtoxidizeroff.Text) 
    PURGEON = Val(txtpurgeon.Text) 
    PURGEOFF = Val(txtpurgeoff.Text) 
    Call SequenceLogicCheck 
    If SequenceCheck = True Then 
        tmrhybrid1.Enabled = True 
    End If 
End Sub 
Private Sub SequenceLogicCheck() 
    If COUNTDOWN > 0 Then 
        If OXIDIZERON > 0 Then 
            If OXIDIZEROFF > OXIDIZERON Then 
                If PURGEON > OXIDIZEROFF Then 
                    If PURGEOFF > PURGEON Then 
                        PURGE = True 
                        SequenceCheck = True 
                    Else 
                        PURGE = False 
                        SequenceCheck = True 



                        txtpurgeon.Text = "---" 
                        txtpurgeoff.Text = "---" 
                    End If 
                Else 
                    PURGE = False 
                    SequenceCheck = True 
                    txtpurgeon.Text = "---" 
                    txtpurgeoff.Text = "---" 
                End If 
            Else 
                txtstatus.Text = "Oxidizer Off must be > Oxidizer On" 
                SequenceCheck = False 
            End If 
        Else 
            txtstatus.Text = "Oxidizer On must occur at T+1 second or later" 
            SequenceCheck = False 
        End If 
    Else 
        txtstatus.Text = "Countdown must be >= 1 second" 
        SequenceCheck = False 
    End If 
End Sub 
 
Private Sub Command1_Click() 
If EZX1.Visible = False Then 
    EZX1.Visible = True 
ElseIf EZX1.Visible = True Then 
    EZX1.Visible = False 
End If 
End Sub 
 
Private Sub Form_Load() 
    tmrhybrid1.Enabled = False 
    tmrhybrid2.Enabled = False 
    cmdstart.Enabled = False 
    txtcount.Text = txtcountdown.Text 
    frmFlexControl.Show 
End Sub 
 
Private Sub tmrhybrid1_Timer() 
        If COUNTDOWNACTIVE = True Then 
            COUNTDOWN = COUNTDOWN - 1 
            txtcount.Text = COUNTDOWN 
            lbltsign.Text = "T minus" 
            txtpurgeoff.BackColor = &H80000005                'white 
            txtoxidizeroff.BackColor = &H80000005             'white 
            txtcountdown.BackColor = &HFF&                    'red 
            Beep 
            If COUNTDOWN <= 1 Then 
                COUNTDOWNACTIVE = False 
            End If 
        ElseIf COUNTDOWNACTIVE = False Then 
            lbltsign.Text = "T plus" 
            txtcount.Text = COUNTUP 
            If COUNTUP < OXIDIZERON Then 
                EZX1.DOKstate = 1                             'Igniter active 
                txtcountdown.BackColor = &H80000005           'white 
                txtignition.BackColor = &HFF&                 'red 
                txtignition.ForeColor = &H80000005 
                txtstatus.Text = "Ignition" 
            ElseIf COUNTUP >= OXIDIZERON And COUNTUP < OXIDIZEROFF Then 
                EZX1.DOKstate = 0 
                txtignition.BackColor = &H80000005            'white 
                txtignition.ForeColor = &HFF&                 'black 
                EZX1.DOSstate = 1                             'Oxidizer ON active 
                txtoxidizeron.BackColor = &HFF&               'red 
                txtstatus.Text = "Oxidizer ON" 
            ElseIf COUNTUP >= OXIDIZEROFF And COUNTUP < PURGEON And PURGE = True Then 
                EZX1.DOSstate = 0 
                txtoxidizeron.BackColor = &H80000005          'white 
                txtoxidizeroff.BackColor = &HFF&              'red 



                EZX1.DOEstate = 1                             'Oxidizer OFF is active 
                txtstatus.Text = "Oxidizer OFF " 
            ElseIf COUNTUP >= PURGEON And COUNTUP < PURGEOFF And PURGE = True Then 
                EZX1.DOEstate = 0 
                EZX1.DONstate = 1                             'Purge Active 
                txtoxidizeroff.BackColor = &H80000005         'white 
                txtpurgeon.BackColor = &HFF&                  'red 
                txtstatus.Text = "Purge ON" 
            ElseIf COUNTUP >= PURGEOFF And PURGE = True Then 
                EZX1.DOSstate = 0 
                txtpurgeon.BackColor = &H80000005             'white 
                txtpurgeoff.BackColor = &HFF&                 'red 
                txtstatus.Text = " Purge OFF" 
                tmrhybrid1.Enabled = False 
            Else 
                EZX1.DOSstate = 0 
                EZX1.DOEstate = 1 
                txtoxidizeron.BackColor = &H80000005          'white 
                txtoxidizeroff.BackColor = &HFF&              'red 
                txtstatus.Text = "Oxidizer OFF" 
                tmrhybrid1.Enabled = False 
                tmrhybrid2.Enabled = True 
            End If 
            EZX1.UpdateCmd = True 
            COUNTUP = COUNTUP + 1 
            Beep 
        End If 
 
End Sub 
 
Private Sub tmrhybrid2_Timer() 
    txtoxidizeroff.BackColor = &H80000005                     'white 
    EZX1.DOEstate = 0 
    EZX1.UpdateCmd = True 
    tmrhybrid2.Enabled = False 
End Sub 
 
Private Sub txtcountdown_Change() 
    txtcount.Text = txtcountdown.Text 
End Sub 
 
Private Sub txtpurgeoff_Change() 
If txtpurgeoff.Text = "0" Then 
    txtpurgeoff.Text = "---" 
End If 
End Sub 
 
Private Sub txtpurgeon_Change() 
If txtpurgeon.Text = "0" Then 
    txtpurgeon.Text = "---" 
End If 
End Sub 


